
Change 1
	Function name
	f_IMS_EmergencyCallSetup_AnnexC22

	Reason for change
	In the case of a normal IMS registration, the IMS PTC will send IMS_CONFIG_REQ(PortsAndSecurityConfig) to IP PTC indicating the unprotected server port in use. This is taken from the contact header of the REGISTER message and used by the IP PTC to send unprotected SIP responses when using UDP.

In the case of IMS emergency call with no registration, the unprotected server port is never set at the IP PTC and the default port 5060 will be used for SIP responses over UDP. This may not match the port on whichy the UE is expecting to receive the responses.



	Summary of change
	On receiving an unprotected INVITE message, call function f_IMS_PortsAndSecurityConfig(), passing false as the 5th parameter indictaing that the UE is not registered (see change 2).

	TTCN module
	IMS_Procedures_CallControl.ttcn


Before change

  function f_IMS_EmergencyCallSetup_AnnexC22(INVITE_A_2_1_Context_Type p_Context := A_2_1_A7,

                                             boolean p_GeolocationInfoAvailable := false,

                                             template (omit) charstring p_OtherCalleeUri := omit) runs on IMS_PTC return INVITE_Request

  { /* p_Context   ..   A_2_1_A6 (INVITE for creating an emergency session in case of no registration) or

     *                  A_2_1_A7 (INVITE for creating an emergency session within an emergency registration) */

    /* @sic R5s130988 change 6/7: sic@

     * in case of A_2_1_A6 there has been no registration beforehand and therefore we don't know whether the UE uses IPv4 or IPv6

     * => we cannot setup any address information before we have received the INVITE

     * => as the SDP message at step 1 contains address information we cannot build up the template in advance but need to match after the INVITE has been received */

    /* @sic R5-140920: additional parameter p_OtherCalleeUri sic@ */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template(value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var template (present) SipUrl v_CalleeContactSipUri;

    var INVITE_Request v_InviteRequest;

    var template (value) MessageHeader v_MessageHeader_InviteResponse_200;

    var template (present) MessageHeader v_MessageHeader_Ack;

    var template (present) SDP_Message v_SDP_Message_Step1;

    var template (value) SDP_Message v_SDP_Message_Step4;

    var SDP_Message v_SDP_Message;

    var boolean v_IsEmergency := true;

    var boolean v_PrackRequiredFor180Ringing := false;  /* => there is no PRACK expected to acknowledge the 180 response */

    var charstring v_FmtAudio;

    // Step 1. Receive INVITE for an Emergency call

    v_IMS_DATA_REQ := f_IMS_INVITE_ReceiveRequest(p_Context, -, p_OtherCalleeUri, -, p_GeolocationInfoAvailable);

    v_InviteRequest := v_IMS_DATA_REQ.Request.Invite;

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    v_SDP_Message_Step1 := f_IMS_BuildSDP_AnnexC22_Step1();   /* @sic R5s130988 change 6/7: has to be done after the INVITE has been received and the address information has been initialised even for A_2_1_A6 sic@ */

    v_SDP_Message := f_IMS_MessageBody_CheckSdpMessageAndGeolocation(v_InviteRequest.messageBody, v_SDP_Message_Step1, p_GeolocationInfoAvailable); /* @sic R5w140112 sic@ sic@ */

    v_FmtAudio := f_SDP_MediaDescr_GetAMR_Any(v_SDP_Message.media_list[0]);      /* any AMR codec is acceptable @sic R5-134795 sic@ */

    if (v_FmtAudio == "") {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "no AMR codec found in given media description");

    }

    // Step 2. SS sends a 100 Trying provisional response.

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine100, f_IMS_InviteResponse_100_MessageHeaderTX(v_InviteRequest))));

    // Step 3. SS sends a 180 Ringing.

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine180, f_IMS_InviteResponse_180_MessageHeaderTX(v_InviteRequest, -, v_PrackRequiredFor180Ringing, v_IsEmergency))));

    // Step 4. SS responds to the invite INVITE with 200 OK including the SDP record.

    v_SDP_Message_Step4 := f_IMS_BuildSDP_AnnexC22_Step4(v_FmtAudio);   /* @sic R5s141244: no origin address to be handed over anymore sic@ */

    v_MessageHeader_InviteResponse_200 := f_IMS_OtherResponse_200_MessageHeaderTX(v_InviteRequest.msgHeader, cs_ContentTypeSDP, tsc_IMS_Emergency_TelUri);

    v_CalleeContactSipUri := f_Addr_Union_GetSipUrl(v_MessageHeader_InviteResponse_200.contact.contactBody.contactAddresses[0].addressField);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_InviteResponse_200, cs_MessageBody_SDP(v_SDP_Message_Step4))));

    // Step 5. Finally, UE acknowledges call setup.

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200, v_IsEmergency);  // @sic R5s130897 change 2 sic@

    IMS_Server.receive(car_IMS_Ack_Request(cr_ACK_Request(v_CalleeContactSipUri, v_MessageHeader_Ack)));

    return v_InviteRequest;

  }        
After change
...

import from IMS_IP_Config all;
...  
  function f_IMS_EmergencyCallSetup_AnnexC22(INVITE_A_2_1_Context_Type p_Context := A_2_1_A7,

                                             boolean p_GeolocationInfoAvailable := false,

                                             template (omit) charstring p_OtherCalleeUri := omit) runs on IMS_PTC return INVITE_Request

  { /* p_Context   ..   A_2_1_A6 (INVITE for creating an emergency session in case of no registration) or

     *                  A_2_1_A7 (INVITE for creating an emergency session within an emergency registration) */

    /* @sic R5s130988 change 6/7: sic@

     * in case of A_2_1_A6 there has been no registration beforehand and therefore we don't know whether the UE uses IPv4 or IPv6

     * => we cannot setup any address information before we have received the INVITE

     * => as the SDP message at step 1 contains address information we cannot build up the template in advance but need to match after the INVITE has been received */

    /* @sic R5-140920: additional parameter p_OtherCalleeUri sic@ */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template(value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var template (present) SipUrl v_CalleeContactSipUri;

    var INVITE_Request v_InviteRequest;

    var template (value) MessageHeader v_MessageHeader_InviteResponse_200;

    var template (present) MessageHeader v_MessageHeader_Ack;

    var template (present) SDP_Message v_SDP_Message_Step1;

    var template (value) SDP_Message v_SDP_Message_Step4;

    var SDP_Message v_SDP_Message;

    var boolean v_IsEmergency := true;

    var boolean v_PrackRequiredFor180Ringing := false;  /* => there is no PRACK expected to acknowledge the 180 response */

    var charstring v_FmtAudio;
    var IP_AddrInfo_Type v_UE_Address := f_IMS_PTC_UE_Address_Get();

    var IP_AddrInfo_Type v_NW_Address := f_IMS_PTC_NW_Address_Get();
    // Step 1. Receive INVITE for an Emergency call

    v_IMS_DATA_REQ := f_IMS_INVITE_ReceiveRequest(p_Context, -, p_OtherCalleeUri, -, p_GeolocationInfoAvailable);

    v_InviteRequest := v_IMS_DATA_REQ.Request.Invite;
    if (v_IMS_DATA_REQ.RoutingInfo.Security == unprotected) {

      v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_InviteRequest.msgHeader);

      f_IMS_PortsAndSecurityConfig(f_SIP_ContactUrl_GetPort(v_ContactUrl),         // may be omit in which case 5060 is chosen by the IP PTC

                                   v_NW_Address,

                                   v_UE_Address, omit, false);

    }
    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    v_SDP_Message_Step1 := f_IMS_BuildSDP_AnnexC22_Step1();   /* @sic R5s130988 change 6/7: has to be done after the INVITE has been received and the address information has been initialised even for A_2_1_A6 sic@ */

    v_SDP_Message := f_IMS_MessageBody_CheckSdpMessageAndGeolocation(v_InviteRequest.messageBody, v_SDP_Message_Step1, p_GeolocationInfoAvailable); /* @sic R5w140112 sic@ sic@ */

    v_FmtAudio := f_SDP_MediaDescr_GetAMR_Any(v_SDP_Message.media_list[0]);      /* any AMR codec is acceptable @sic R5-134795 sic@ */

    if (v_FmtAudio == "") {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "no AMR codec found in given media description");

    }

    // Step 2. SS sends a 100 Trying provisional response.

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine100, f_IMS_InviteResponse_100_MessageHeaderTX(v_InviteRequest))));

    // Step 3. SS sends a 180 Ringing.

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine180, f_IMS_InviteResponse_180_MessageHeaderTX(v_InviteRequest, -, v_PrackRequiredFor180Ringing, v_IsEmergency))));

    // Step 4. SS responds to the invite INVITE with 200 OK including the SDP record.

    v_SDP_Message_Step4 := f_IMS_BuildSDP_AnnexC22_Step4(v_FmtAudio);   /* @sic R5s141244: no origin address to be handed over anymore sic@ */

    v_MessageHeader_InviteResponse_200 := f_IMS_OtherResponse_200_MessageHeaderTX(v_InviteRequest.msgHeader, cs_ContentTypeSDP, tsc_IMS_Emergency_TelUri);

    v_CalleeContactSipUri := f_Addr_Union_GetSipUrl(v_MessageHeader_InviteResponse_200.contact.contactBody.contactAddresses[0].addressField);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_InviteResponse_200, cs_MessageBody_SDP(v_SDP_Message_Step4))));

    // Step 5. Finally, UE acknowledges call setup.

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200, v_IsEmergency);  // @sic R5s130897 change 2 sic@

    IMS_Server.receive(car_IMS_Ack_Request(cr_ACK_Request(v_CalleeContactSipUri, v_MessageHeader_Ack)));

    return v_InviteRequest;

  }
Change 2
	Function name
	f_IMS_PortsAndSecurityConfig()

	Reason for change
	In the case of a normal IMS registration, the IMS PTC will send IMS_CONFIG_REQ(PortsAndSecurityConfig) to IP PTC indicating the unprotected server port in use. This is taken from the contact header of the REGISTER message and used by the IP PTC to send unprotected SIP responses when using UDP.

In the case of IMS emergency call with no registration, the unprotected server port is never set at the IP PTC and the default port 5060 will be used for SIP responses over UDP. This may not match the port on whichy the UE is expecting to receive the responses.



	Summary of change
	Add a new Boolean parameter indicating whether the UE is registered. If this is set to true then “reigstrationInfo” should be omitted.

	TTCN module
	IMS_IP_Config.ttcn


Before change

  function f_IMS_PortsAndSecurityConfig(template (omit)  PortNumber_Type p_UnprotectedPort_us,

                                        IP_AddrInfo_Type p_NW_Address,

                                        IP_AddrInfo_Type p_UE_Address,

                                        template (omit)  IMS_SecurityInfo_Type p_SecurityInfo := omit) runs on IMS_PTC

  { /* configure IMS security; the IP PTC returns SPIs which it has assigned;

     * p_SecurityInfo is omit in case of GIBA

     * @sic R5s130266 change 1.3, 4, 5 - MCC160 Implementation: SPIs for re-authentication sic@ */

    if (isvalue(p_SecurityInfo)) {

      if (valueof(p_SecurityInfo.SPIs.SPI_us) == valueof(p_SecurityInfo.SPIs.SPI_uc)) {

        FatalError(__FILE__, __LINE__, "SPI_us and SPI_uc shall have different values");

      }

    }

    fl_IMS_ConfigureIP(cas_IMS_PortsAndSecurityConfig_REQ(cs_IMS_PortsAndSecurityConfig(p_UnprotectedPort_us, cs_IMS_RegistrationInfo(p_NW_Address, p_UE_Address, p_SecurityInfo))),

                       car_IMS_PortsAndSecurityConfig_CNF);  /* @sic R5s140780 - MCC160 Implementation sic@ */

  }                                               
After change
  function f_IMS_PortsAndSecurityConfig(template (omit)  PortNumber_Type p_UnprotectedPort_us,

                                        IP_AddrInfo_Type p_NW_Address,

                                        IP_AddrInfo_Type p_UE_Address,

                                        template (omit)  IMS_SecurityInfo_Type p_SecurityInfo := omit,

                                        boolean p_Registered := true) runs on IMS_PTC

  { /* configure IMS security; the IP PTC returns SPIs which it has assigned;

     * p_SecurityInfo is omit in case of GIBA

     * @sic R5s130266 change 1.3, 4, 5 - MCC160 Implementation: SPIs for re-authentication sic@ */

    var IMS_RegistrationInfo_Type v_RegistrationInfo := omit;
    if (isvalue(p_SecurityInfo)) {

      if (valueof(p_SecurityInfo.SPIs.SPI_us) == valueof(p_SecurityInfo.SPIs.SPI_uc)) {

        FatalError(__FILE__, __LINE__, "SPI_us and SPI_uc shall have different values");

      }

    }

    if (p_Registered)

    {

      v_RegistrationInfo := cs_IMS_RegistrationInfo(p_NW_Address, p_UE_Address, p_SecurityInfo);

    }
    fl_IMS_ConfigureIP(cas_IMS_PortsAndSecurityConfig_REQ(cs_IMS_PortsAndSecurityConfig(p_UnprotectedPort_us, v_RegistrationInfo)),

                       car_IMS_PortsAndSecurityConfig_CNF);  /* @sic R5s140780 - MCC160 Implementation sic@ */

  }
Change 3
	Function name
	Type IMS_PortsAndSecurityConfigReq_Type

	Reason for change
	In the case of a normal IMS registration, the IMS PTC will send IMS_CONFIG_REQ(PortsAndSecurityConfig) to IP PTC indicating the unprotected server port in use. This is taken from the contact header of the REGISTER message and used by the IP PTC to send unprotected SIP responses when using UDP.

In the case of IMS emergency call with no registration, the unprotected server port is never set at the IP PTC and the default port 5060 will be used for SIP responses over UDP. This may not match the port on whichy the UE is expecting to receive the responses.



	Summary of change
	Make field “RegistrationInfo” optional to allow this message to be used to set the unprotected server port in the case of IMS emergency call with no registration

	TTCN module
	IMS_ASP_TypeDefs.ttcn


Before change

  type record IMS_PortsAndSecurityConfigReq_Type {

    PortNumber_Type                     UnprotectedPort_us  optional,   // 5060 per default

    IMS_RegistrationInfo_Type           RegistrationInfo

  };

After change
    type record IMS_PortsAndSecurityConfigReq_Type {

    PortNumber_Type                     UnprotectedPort_us  optional,   // 5060 per default

    IMS_RegistrationInfo_Type           RegistrationInfo optional
  };

Change 4
	Function name
	Altstep a_IP_IMS_Config()

	Reason for change
	In the case of a normal IMS registration, the IMS PTC will send IMS_CONFIG_REQ(PortsAndSecurityConfig) to IP PTC indicating the unprotected server port in use. This is taken from the contact header of the REGISTER message and used by the IP PTC to send unprotected SIP responses when using UDP.

In the case of IMS emergency call with no registration, the unprotected server port is never set at the IP PTC and the default port 5060 will be used for SIP responses over UDP. This may not match the port on whichy the UE is expecting to receive the responses.



	Summary of change
	Update handling of IMS_CONFIG_REQ(PortsAndSecurityConfig) to allow for “reigstrationInfo” being omitted.

	TTCN module
	IP_PTC_IMS_Handler.ttcn


Before change

      altstep a_IP_IMS_Config(inout IMS_Server_Type p_ImsServer,

                          IP_IMS_CTRL_PORT p_Port) runs on IP_PTC

  { /* @sic R5s130266 change 1.3, 4, 5 - MCC160 Implementation: SPIs and protected ports are fully controlled by the IMS PTC sic@ */

    /* @sic R5s130900 - MCC160 implementation for re-authentication sic@ */

    var IMS_CONFIG_REQ v_IMS_CONFIG_REQ;

    var IMS_PortsAndSecurityConfigReq_Type v_PortsAndSecurityConfig;

    var IMS_RegistrationInfo_Type v_RegistrationInfo;

    var boolean v_WaitForCNF;

    [] p_Port.receive(car_IMS_InstallKey_REQ) -> value v_IMS_CONFIG_REQ

      {

        p_ImsServer.SecurityKeys := v_IMS_CONFIG_REQ.InstallKey;

        p_Port.send(cas_IMS_InstallKey_CNF);

      }

    [] p_Port.receive(car_IMS_PortsAndSecurityConfig_REQ) -> value v_IMS_CONFIG_REQ

      {

        v_PortsAndSecurityConfig := v_IMS_CONFIG_REQ.PortsAndSecurityConfig;

        v_RegistrationInfo := v_PortsAndSecurityConfig.RegistrationInfo;

        if (not isvalue(p_ImsServer.RegistrationAddress)) {

          p_ImsServer.RegistrationAddress.NW := v_RegistrationInfo.NW_Address;

          p_ImsServer.RegistrationAddress.UE := v_RegistrationInfo.UE_Address;

        }

        if (isvalue(v_PortsAndSecurityConfig.UnprotectedPort_us)) {

          p_ImsServer.UnprotectedContext.SecurityInfo.Unprotected.Port_us := v_PortsAndSecurityConfig.UnprotectedPort_us;   /* set UE's port number as requested by the UE */

        } else {

          p_ImsServer.UnprotectedContext.SecurityInfo.Unprotected.Port_us := tsc_IMS_PortNumber_5060;                       /* restore port number e.g. if it has been set to other value in previous session */

        }

        if (isvalue(v_RegistrationInfo.SecurityInfo)) {                                                                     /* non-GIBA   @sic R5s120858 change 6..8 - MCC160 implementation sic@ */

          if (not isvalue(p_ImsServer.SecurityKeys)) {

            FatalError(__FILE__, __LINE__, "security keys missing");

          }

          fl_IMS_Server_AddSecurityContext(p_ImsServer, v_RegistrationInfo);                                                /* @sic R5-145732: support of 2nd security context sic@ */

        }

        else {

          fl_IMS_Server_RemoveSecurityContext(p_ImsServer);

        }

        p_Port.send(cas_IMS_PortsAndSecurityConfig_CNF);

      }

    [] p_Port.receive(car_IMS_SecurityRelease_REQ) -> value v_IMS_CONFIG_REQ

      {

        v_WaitForCNF := false;

        fl_IMS_Server_RemoveSecurityContext(p_ImsServer, v_WaitForCNF);      /* @sic R5-145732: support of 2nd security context sic@ */

        p_Port.send(cas_IMS_SecurityRelease_CNF);

      }

    [] p_Port.receive(car_IMS_RegInfoRelease_REQ) -> value v_IMS_CONFIG_REQ  /* @sic R5s140123 change 8 sic@ */

      {

        v_WaitForCNF := false;

        fl_IMS_Server_RemoveSecurityContext(p_ImsServer, v_WaitForCNF);      /* @sic R5s141154 sic@ */

        p_ImsServer.RegistrationAddress := omit;

        p_Port.send(cas_IMS_RegInfoRelease_CNF);

      }

    [] p_Port.receive(car_IMS_CloseTCP_REQ) -> value v_IMS_CONFIG_REQ        /* @sic R5s130988 change 1 - MCC160 implementation sic@ */

      {

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, unprotected);

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext1);

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext2);

        p_Port.send(cas_IMS_CloseTCP_CNF);

      }

  }

After change
  altstep a_IP_IMS_Config(inout IMS_Server_Type p_ImsServer,

                          IP_IMS_CTRL_PORT p_Port) runs on IP_PTC

  { /* @sic R5s130266 change 1.3, 4, 5 - MCC160 Implementation: SPIs and protected ports are fully controlled by the IMS PTC sic@ */

    /* @sic R5s130900 - MCC160 implementation for re-authentication sic@ */

    var IMS_CONFIG_REQ v_IMS_CONFIG_REQ;

    var IMS_PortsAndSecurityConfigReq_Type v_PortsAndSecurityConfig;

    var template (omit) IMS_RegistrationInfo_Type v_RegistrationInfo;

    var boolean v_WaitForCNF;

    [] p_Port.receive(car_IMS_InstallKey_REQ) -> value v_IMS_CONFIG_REQ

      {

        p_ImsServer.SecurityKeys := v_IMS_CONFIG_REQ.InstallKey;

        p_Port.send(cas_IMS_InstallKey_CNF);

      }

    [] p_Port.receive(car_IMS_PortsAndSecurityConfig_REQ) -> value v_IMS_CONFIG_REQ

      {

        v_PortsAndSecurityConfig := v_IMS_CONFIG_REQ.PortsAndSecurityConfig;

        v_RegistrationInfo := v_PortsAndSecurityConfig.RegistrationInfo;

        if (not isvalue(p_ImsServer.RegistrationAddress) and isvalue (v_RegistrationInfo)) {

          p_ImsServer.RegistrationAddress.NW := v_RegistrationInfo.NW_Address;

          p_ImsServer.RegistrationAddress.UE := v_RegistrationInfo.UE_Address;

        }

        if (isvalue(v_PortsAndSecurityConfig.UnprotectedPort_us)) {

          p_ImsServer.UnprotectedContext.SecurityInfo.Unprotected.Port_us := v_PortsAndSecurityConfig.UnprotectedPort_us;   /* set UE's port number as requested by the UE */

        } else {

          p_ImsServer.UnprotectedContext.SecurityInfo.Unprotected.Port_us := tsc_IMS_PortNumber_5060;                       /* restore port number e.g. if it has been set to other value in previous session */

        }

        if (isvalue(v_RegistrationInfo) and isvalue(v_RegistrationInfo.SecurityInfo)) {                                                                     /* non-GIBA   @sic R5s120858 change 6..8 - MCC160 implementation sic@ */

          if (not isvalue(p_ImsServer.SecurityKeys)) {

            FatalError(__FILE__, __LINE__, "security keys missing");

          }

          fl_IMS_Server_AddSecurityContext(p_ImsServer, v_RegistrationInfo);                                                /* @sic R5-145732: support of 2nd security context sic@ */

        }

        else {

          fl_IMS_Server_RemoveSecurityContext(p_ImsServer);

        }

        p_Port.send(cas_IMS_PortsAndSecurityConfig_CNF);

      }

    [] p_Port.receive(car_IMS_SecurityRelease_REQ) -> value v_IMS_CONFIG_REQ

      {

        v_WaitForCNF := false;

        fl_IMS_Server_RemoveSecurityContext(p_ImsServer, v_WaitForCNF);      /* @sic R5-145732: support of 2nd security context sic@ */

        p_Port.send(cas_IMS_SecurityRelease_CNF);

      }

    [] p_Port.receive(car_IMS_RegInfoRelease_REQ) -> value v_IMS_CONFIG_REQ  /* @sic R5s140123 change 8 sic@ */

      {

        v_WaitForCNF := false;

        fl_IMS_Server_RemoveSecurityContext(p_ImsServer, v_WaitForCNF);      /* @sic R5s141154 sic@ */

        p_ImsServer.RegistrationAddress := omit;

        p_Port.send(cas_IMS_RegInfoRelease_CNF);

      }

    [] p_Port.receive(car_IMS_CloseTCP_REQ) -> value v_IMS_CONFIG_REQ        /* @sic R5s130988 change 1 - MCC160 implementation sic@ */

      {

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, unprotected);

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext1);

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext2);

        p_Port.send(cas_IMS_CloseTCP_CNF);

      }

  }

